Double-carrier FM

The use of two carrier oscillators can be employed to generate a fixed formant region in the spectrum. As formants are very important for instrumental and vocal sounds, this technique can be useful in the simulation of certain instruments.

The idea here is that two carriers, with different frequencies are used. The second carrier will have a frequency that is close to the centre of the formant region, so that it generates side-bands around it. The index of modulation for this carrier will be generally small (a fraction of the index for the first carrier frequency), so that the energy can be concentrated at the carrier frequency (which is at the centre of the formant region). Also, the amplitude of that carrier is adjusted to control the relative strength of the formant.

The second carrier frequency is chosen to be a harmonic of the fundamental frequency that is closest to the desired formant region, by the following expression:

$$f_{c2} = \mathrm{int}[(\frac{f_f}{f_0} + 0.5)] \times f_0$$

where f_f is the formant frequency and f_0 is the fundamental frequency, the expression int[.] returns an integral value, with the decimal part discarded. So the formula gives us the nearest harmonic to the formant centre frequency. The second carrier frequency will always be a harmonic.

A straightforward application of this design is shown in the csound code below:

```
indx2 =  p8            ;mod index for carrier 2
indx1 = p7             ;mod index for carrier 1
iff   = p6             ;formant frequency
if0 = p5               ;fundamental
ifm = if0              ;FC1 = FM = fund (1:1 ration)
ifc1 = if0
ifc2 = int((iff/if0) + 0.5) * if0 ; FC2
iamp = p4              ;carrier 1 amp
iampfac =  p9          ;carrier 2 amp factor
iratio = indx2/indx1   ;index ratio to scale mod signal for carrier 2


amod oscili  indx1, ifm, 1
acar1 oscili  iamp, amod+ifc1, 1
acar2 oscili  iamp*iampfac, (amod * iratio) + ifc2, 1
          out  acar1+acar2
```

Of course, this is the ‘bare-bones’ design. To make things more realistic, we will have to add things like amplitude envelopes, vibrato, etc.. John Chowning has designed an instrument that uses this principle to generate a soprano voice sound, with vibrato and envelopes to complement it.

Here’s the csound code for it:

```
; soprano FM, Dodge & Jerse, pp 120-1 : orchestra
; Victor Lazzarini, 1998

sr= 44100
kr = 4410
ksmps = 10
nchnls= 1

instr 1

idur = p3
iamp = p4
ifp  = cpspch(p5)            ; carrier freq
ifm =  ifp                   ; mod freq
isine  =  1

;   tables
ipowfp pow    2, ((ifp - 196)/1372) - 1  ; range G3 (196Hz) - G6 (1568Hz)
iff    tablei  ipowfp*ftlen(3), 3
i1     tablei  ipowfp*ftlen(5), 5
i2     tablei  ipowfp*ftlen(6), 6
ia2    tablei  ipowfp*ftlen(4), 4


;   vibrato
ivibwth =.0002*1.4427*log(ifp)
ivibrate = 5.8
kvib   linen ivibwth, .6, p3, .1
kvibr  randi ivibwth, 125
kvibra oscil kvib, ivibrate, isine
kvibo  oscil .01, 1/p3, 2
;kvg linseg  0, p3/2, 1, p3/2, 1
kv     =  (kvibr+kvibra+kvibo+1);

; modulator
amod    oscil    i1*ifm, ifm, isine
amod2    =       (i2/i1)*amod

; carriers
iamp1  =        ia2*(sqrt(iamp*iamp*iamp))
kamp   linen    sqrt(iamp)*(1-ia2),.1, p3, .08
acar1  oscil    kamp, (amod+ifp)*kv, isine
kamp2  linseg   0,.1, 0, .5, iamp1, p3-.6, iamp1, .08, 0
acar2  oscil    kamp2, ((int((iff/ifp) + .5)*ifp)+amod2)*kv, isine

         out    (acar1 +acar2)*23400

endin


;score for Chowning's soprano voice

f1 0 1024  10  1 ; senoide
f2 0 1024   7  0  25 1 25 .75 924 .75   ; port

f3 0  256   -7  2800 42 2800 84 2400 86 2000 44 2000; f2
f4 0  256   -7  .04 42 .02 84 .05 130 .75; i1
f5 0  256   -7  .25 84 .125 172 .075; A2
f6 0  256   -7  5 42 2.5 42 1.25 130 4 42 3 ; i2


;       dur amp (0 < amp < 1)   pitch
i1 0    5   .4                 8.00
i1 +    5   .5                 9.02
i1 +    5   .6                 8.09
i1 +    5   .65                8.11
i1 +    5   .78                9.00
```

Multiple Carriers

Extending Chowning's soprano instrument, here is a 6-carrier design, which uses tables to store formant centre frequencies and amplitudes. These tables are scanned cyclically using oscillators and they provide the values to be used in the calculation of the carrier frequencies and their amplitudes. The amplitudes are stored as dB values, which then are translated using **ampdb(.)**. The 5 extra carriers model 5 formant regions. The vibrato is added to each carrier (as in the Chowning example).

```
instr 1

i5 =    p5           ; fundamental
imax =  p6           ; imax
indx =   imax        ; carrier 0 freq
indx1 =  1         ; 1 , etc
indx2 =  1
indx3 =  1
indx4 =  1
indx5 =  1

kndx    linen   indx, .01, p3, .5  ; index envelope
kenv    linen   p4, .1, p3, .5     ; amplitude envelope

ivibwth =  .2*1.4427*log(i5)           ; vibrato
ivibrate = 6.8
kvib   linen ivibwth, .6, p3, .1
kvibr  randi ivibwth, 125
kvibra oscil kvib, ivibrate, 1
kv     =   kvibr+kvibra

kf1   oscil 1, 1/p3, 5             ; formant freq
kf2   oscil 1, 1/p3, 6
kf3   oscil 1, 1/p3, 7
kf4   oscil 1, 1/p3, 8
kf5   oscil 1, 1/p3, 9
                                   ; carrier amplitudes
ka2  oscili 1, 1/p3, 11
ka3  oscili 1, 1/p3, 12
ka4  oscili 1, 1/p3, 13
ka5  oscili 1, 1/p3, 14

kamp2 = ampdb(ka2)
kamp3 = ampdb(ka3)
kamp4 = ampdb(ka4)
kamp5 = ampdb(ka5)

kscal = 1/(1+kamp2+kamp3+kamp4+kamp5) ; used to scale output
                                      ; to maximum amp

amod oscil kndx*i5, i5, 1          ; modulator

amod1 = amod*(indx1/indx)          ; modulation signals
amod2 = amod*(indx2/indx)
amod3 = amod*(indx3/indx)
amod4 = amod*(indx4/indx)
amod5 = amod*(indx5/indx)
                                   ; carrier
aca   oscil kenv, i5+amod+kv, 1
aca1  oscil kenv, (int(kf1/i5 +.5)*i5)+amod1+kv  ,1
aca2  oscil kenv*kamp2, (int(kf2/i5 +.5)*i5)+amod2+kv  ,1
aca3  oscil kenv*kamp3, (int(kf3/i5 +.5)*i5)+amod3+kv  ,1
aca4  oscil kenv*kamp4, (int(kf4/i5 +.5)*i5)+amod4+kv  ,1
aca5  oscil kenv*kamp5, (int(kf5/i5 +.5)*i5)+amod5+kv  ,1
```

```
        out (aca1+aca2+aca3+aca4+aca5+aca)*kscal
endin

;score
f1 0 1024  10  1 ; sine

f5 0 8  -2 400   604  325  360  360 325 604 409 ; formants (eh ah o u
                                                ; u o ah eh)
f6 0 8  -2 1700 1000  700  750  750 700 1000 1700
f7 0 8  -2 2300 2450 2550 2400  2400 2550 2450 2300
f8 0 8  -2 2900 2700 2850 2675  2675 2850 2700 2900
f9 0 8  -2 3400 3240 3100  2950 2950 3100 3240 3400

f11 0 8  -2 -9  -6  -12  -12  -12  -12 -6  -9 ; amps (dB)
f12 0 8  -2 -8  -12 -26  -29  -29 -26 -12 -8
f13 0 8  -2 -11 -11 -22  -26  -26 -22 -11 -11
f14 0 8  -2 -19 -24 -28 -35  -35 -28 -24 -19

;               fun  imax
i1 0 .3  8000  150  .1
i1 + .5  8000  <   <
i1 + .7  8000  <  <
i1 + .9  8000  <   <
i1 + 1.3  8000  <  <
i1 + 1.9  8000  <   <
i1 + 2.5  8000  <  <
i1 + 3.3  8000  <  <
i1 + 4.9  8000  <   <
i1 + 6.5  8000   < <
i1 + 10   8000  155  1.5
```

Complex Modulation

Here’s an example, by Bill Schoestaedt, of complex modulation using three modulators, creating a cello-like sound. The modulating frequencies are 1,3 and 4 times the carrier frequency (also the fundamental). The indices are dependent on the fundamental and a little extra is added (by means of a timbral envelope) at the attack period. There is a little vibrato and some band-limited noise around 2KHz is added to the sound only at the attack period (simulating bow noise).

```
instr 1

idur = p3
iamp = p4
ifc = cpspch(p5)
ifm1 = ifc
ifm2 = 3*ifc
ifm3 = 4*ifc

; max indices of mod
i1 = 7.5/log(ifc)
i2 = 15/sqrt(ifc)
i3 = 1.25/sqrt(ifc)

;timbral envelope
ksig1   linseg   1, .2, 0, idur-.2, 0

;modulators
```

```
afm1    oscil (i1+ksig1)*ifm1, ifm1, 1
afm2    oscil (i2+ksig1)*ifm2, ifm2, 1
afm3    oscil (i3+ksig1)*ifm3, ifm3, 1

afm      =  afm1 + afm2 + afm3 + ifc

; vibrato
kran    randi .0075, 15
kvib    oscil kran, 5.5, 1

acar    oscil iamp, (kvib+1)*afm, 1

; attack noise: band-limited noise
; ring modulated by a 2KHz sine wave

atta    randi ksig1*iamp, .2*ifc
attnoi  oscil atta, 2000, 1

aout    linen attnoi+acar, .2, idur, .2
        out  aout
endin
```

;score

```
f1 0 1024 10  1

i1 0 2 16000 7.00
i1 1.95 2 10000 6.10
i1 3.95 3 16000  6.09
```

Phase Modulation

Here's an example of simple PM, translated from the basic FM instrument. Can you supply a score for it?

```
instr 1

idur = p3   ; total duration
iamp = p4   ; max amplitude
ifc = p5    ; carrier freq
ifm = p6    ; modulator freq
indx = p7   ; max index
iatt = p8   ; attack time
isin = 1    ; sine table
i2pi = 1/6.2832 ; 1/2PI

; amp & indx envelopes
kamp  linen  iamp, iatt, idur, iatt*2
kndx  linen  indx*i2pi, iatt*5, idur, iatt*4

; modulator
amod  oscili kndx, ifm, isin

; carrier phase signal
aphs  phasor ifc
; carrier table lookup
acar  tablei amod+aphs, isin, 1, 0, 1

      out   kamp*acar ; output, envelope-shaped

endin
```